

# JAsCo for Linking Business Rules to Object-Oriented Software

## ABSTRACT

Object-oriented software applications that support a particular business or domain consist of substantial core application functionality and *business rules*. Since business rules tend to evolve frequently, it is important to *separate* them from the core application. However, current approaches that support business rules at the implementation level only separate the business rules themselves and not the code that *links* them to the core application. We observe that this code *crosscuts* the core application. As a result, Aspect-Oriented Programming is required to separate and encapsulate the linking code. In addition to this, we identify several other requirements for obtaining highly flexible and configurable business rules. In previous work we conducted an experiment with *AspectJ* for separating the business rule links. Although this delivered satisfying results for some of the requirements, many were not satisfied. This paper shows how JAsCo, an aspect-oriented implementation language combining the advantages of AspectJ's expressiveness with plug-and-play characteristics of components, succeeds in fulfilling the remaining requirements.

**KEYWORDS:** Object-Oriented Software Engineering, Business Rules, Aspect-Oriented Programming

## 1 INTRODUCTION

Software that supports and manages business domains and processes – such as found in electronic commerce, the financial and legal fields, television and radio broadcasting – comes in a wide variety: information systems that are inherently data-oriented [10], rule systems that automate knowledge-intensive domains [24], and software that has a substantial core application functionality supporting the user in his or her tasks without fully automating them. In this paper we focus on the latter kind of software applications, developed using object-oriented or component-based software development techniques.

In this context it is increasingly important to consider *business rules* as a means to capture some business policies explicitly. The Business Rules Group defines a business rule as *a statement that defines or constraints some aspect of the business. It is intended to assert business structure or to control the behaviour of the business* [5]. Business rules tend to evolve more frequently than the core application functionality

[15][1][28]. Therefore, it is crucial to *separate* business rules from the core application, in order to *trace* them to business policies and decisions, *externalise* them for a business audience, and *change* them [28]. A business rule is applied at an event, which is a well-defined point in the execution of the core application functionality.

However, approaches that advocate and support the separation of business rules at the implementation level, fail to separate and encapsulate the code that *links* the business rules to the core application. One has to adapt the source code of the core application manually at different places each time business rules change. This phenomenon is known as *crosscutting* code in the area of *Aspect-Oriented Programming* (AOP) [3][8]. AOP advocates extending standard modularisation constructs of a programming language with additional constructs to encapsulate crosscutting code. Although AOP is usually employed for encapsulating implementation-level issues like logging and synchronisation, we introduce the idea of domain knowledge as an aspect in [12] and [11]. In [7] and [6] we conducted an experiment which uses *AspectJ* [16] for encapsulating the crosscutting business rule links.

However, separating and encapsulating the business rule links is not sufficient in order to achieve highly flexible and configurable business rules. We identify other requirements, which are presented in the next section. AspectJ addresses some of these issues successfully because of its expressiveness with respect to describing and manipulating events in the core application. Some other requirements however, are not adequately satisfied. This paper reports on our efforts to meet the requirements for the business rule links using *JAsCo* [26], which is an aspect-oriented implementation language integrating the ideas of AOP into Component-Based Software Development (CBSD) [25]. JAsCo combines the advantages of AspectJ's expressiveness with the idea of fully reusable and highly configurable plug-and-play characteristics of components.

After introducing AOP and our requirements in Section 2, we discuss the JAsCo language in Section 3. We show how JAsCo fulfils the requirements for linking business rules in Section 4. Business rules for price personalisation in e-commerce are used as a running example throughout the paper since personalisation is an increasingly important issue [9] and e-commerce is a favoured case of most business rules approaches [14][2][22]. Finally, we discuss related work in Section 5 and conclude in Section 6.

## **2 ASPECT-ORIENTED PROGRAMMING FOR BUSINESS RULES**

### **2.1 Introduction to Aspect-Oriented Programming**

Aspect Oriented Programming (AOP) argues that some concerns of a system, such as synchronisation

and logging, cannot be cleanly modularized using current software engineering methodologies as they are scattered all over the different modules of the system. Similar logic is thus repeated in different modules. Due to this code duplication, it becomes very hard to add, edit and remove such a *crosscutting* aspect in the system. The ultimate goal of AOP is to achieve a better separation of concerns. To this end, AOP approaches introduce a new concept that is able to modularize crosscutting concerns, called an *aspect*. An aspect defines a set of *join points* in the target application where the normal execution is altered. Aspect *weavers* are used to weave the aspect logic into the target application.

Nowadays, several AOP approaches, such as AspectJ, *Composition Filters* [4], *HyperJ* [19][27] and *DemeterJ* [18] are available. These technologies have already been applied on large industrial projects by for instance Boeing, IBM and Verizon Communications. For more information about AOP in general, we refer to [3] and [8].

## 2.2 Requirements for Business Rule Links

Business rule logic can be seen as the combination of the business rules themselves and the specification of the business rule link with core application events. Ideally, business rules are represented as an “*if condition then action*”-statement. However, at the implementation level of object-oriented applications business rules are typically modelled as classes [1][21]. A business rule class defines operations for the condition and the action. Hence, no aspect-oriented support is needed for the business rules themselves, because object-oriented techniques suffice for encapsulating and reusing them.

However, not only the reusability of business rules is required. The business rule links should also be encapsulated in order to enable reusability. As business rule links crosscut the core application, AOP techniques are required.

Moreover, we identify a set of requirements that should be satisfied for a particular AOP approach to be suitable [7]:

- ✓ connect business rules to core application events which depend on run-time properties,
- ✓ pass necessary business objects to an event in order to make business rules applicable at that event,
- ✓ reuse a business rule link at different events,
- ✓ combine, prioritize and exclude business rules when they interfere with one another,
- ✓ control the instantiation, initialisation and execution of business rule links,
- ✓ and preferably accomplish the above dynamically without interrupting the application execution.

AspectJ, which is able to describe and manipulate events in a very expressive way, fulfils the first two requirements successfully, whereas the other requirements are only met partially or not at all [7]. The next section introduces JAsCo, after which we show in Section 4 how it addresses the last four requirements successfully.

## 3 JASCO

### 3.1 Introduction to JAsCo

The JAsCo language is primarily based upon two existing AOP approaches: AspectJ and *Aspectual Components* [17]. AspectJ's main advantage is the expressiveness of its language to describe join points. However, AspectJ aspects are not reusable, since the context on which an aspect needs to be deployed is specified directly in the aspect definition. To overcome this problem, Karl Lieberherr et al introduce the concept of Aspectual Components. They claim that doing AOP means being able to express each aspect separately, in terms of its own modular structure. Using this model, an aspect is described as a set of abstract join points which are resolved when an aspect is combined with the base modules of a software system. This way, the aspect behaviour is kept separate from the base components, even at run time. JAsCo combines the expressive power of AspectJ with the aspect independency idea of Aspectual Components.

Originally JAsCo was designed to integrate aspect-oriented ideas into Component-Based Software Development. However, JAsCo has some characteristics that are also useful in an object-oriented context:

- ✓ Aspects are described independent of a concrete context, making them highly reusable.
- ✓ JAsCo allows easy application and removal of aspects at run time.
- ✓ JAsCo has extensive support for specifying aspect combinations.

The JAsCo language itself stays as close as possible to the regular Java syntax and introduces two new concepts: *Aspect Beans* and *Connectors*. An Aspect Bean is an extension of the Java Bean component that is able to specify crosscutting behaviour. A Connector on the other hand is responsible for applying the crosscutting behaviour of the Aspect Beans and for declaring how several of these aspects collaborate. On a technical level we introduce a new, backward compatible component model that enables run-time application and removal of Connectors. The next two sections explain Aspect Beans and Connectors in more detail. For more information about JAsCo and the JAsCo component model, we refer to [26]. Notice that although JAsCo is designed for component-based applications, it is also possible to employ JAsCo in an object-oriented context.

## 3.2 Aspect Beans

Aspects Beans describe some behaviour that would normally crosscut several parts of a system. An Aspect Bean is an extended version of a regular Java Bean that defines one or more logically related hooks as a special kind of inner classes. The Aspect Bean itself is used to implement the business rule and to specify the Hooks that are used to describe the linking of the rule with the core application. Hence, an Aspect Bean is able to combine the two parts of the business rule logic in the same module, but is still able to maintain the desired separation and independence between the business rule and the linking.

```
1  abstract class BRPriceDiscount {
2      private Float discount;
3      public void setDiscount(Float aDiscount) {
4          discount = aDiscount;
5      }
6      public Float getDiscount() {
7          return discount;
8      }
9      abstract public boolean discountCondition(Customer);
10     public Float applyDiscount(Float aPrice) {
11         float price = aPrice.floatValue();
12         return new Float(price - (price * getDiscount()));
13     }
14
15     hook BRPriceDiscountHook {
16         BRPriceDiscountHook(Float method(Customer aCustomer)) {
17             execute(method);
18         }
19         isApplicable() {
20             return discountCondition(aCustomer);
21         }
22         replace() {
23             Float price = method(aCustomer);
24             return applyDiscount(price);
25         }
26     }
27 }
```

**Figure 1: The implementation of the abstract business rule**

Figure 1 illustrates an abstract discount Aspect Bean from which all discount business rules inherit. The `BRPriceDiscount`-Aspect Bean describes the business rule (lines 2 to 13) and declares a `BRPriceDiscount`-hook (lines 15 to 26) that describes the linking of the business rule with the core application. A hook specifies **when** the normal execution of the base program should be interrupted, and **what** extra behaviour should be executed at that precise moment in time. In order to define when the functionality of a hook should be executed, the hook is equipped with at least one constructor (lines 16 to 18) that takes one or more *abstract method parameters* as input. These abstract method parameters are used for describing the

abstract context of a hook. This generic specification of the context of an aspect makes business rules reusable and as a result deployable in different contexts. The `BRPriceDiscount`-hook specifies that its behaviour is deployable on every method that takes a *Customer* as input and that returns a *Float*-value. The constructor body describes how the join points of a hook initialisation should be computed. In this particular case, the constructor-body (line 17) specifies that the functionality of the `BRPriceDiscount`-hook should be performed whenever *method* is executed. The behaviour methods of a hook on the other hand, are used for specifying the various actions a hook needs to perform whenever one of its calculated join points is encountered. Three behaviour methods are available: *before*, *after* and *replace*. The *replace* behaviour method of the `BRPriceDiscount`-hook (lines 22 to 25) specifies that some discount is given, whenever the `isApplicable`-method returns true. The `isApplicable`-method specifies a dynamic condition that is executed at Connector, to check whether the behaviour of an aspect should be executed. The specific discount-percentage and the `discountCondition`-method are undetermined at the moment, because this information is specific to each business rule that extends the abstract `BRPriceDiscount`-Aspect Bean.

```
1  class ChristmasBR extends BRPriceDiscount {
2      public boolean discountCondition(Customer customer) {
3          //return true if Christmas
4      }
5  }
```

**Figure 2: The Christmas business rule**

Figure 2 illustrates the Christmas business rule, which is a concrete implementation of the abstract discount business rule presented in Figure 1. The `ChristmasBR`-rule only implements the `discountCondition`-method, since the logic behind this method is specific for each discount business rule. In this particular case, the `discountCondition`-method returns true if it is Christmas. As the `ChristmasBR`-rule extends the abstract discount Aspect Bean, it also inherits the `BRPriceDiscount`-hook.

### 3.3 Connectors

Connectors are used for instantiating one or more logically related hooks with a specific context (method or event signatures) and for specifying advanced aspect-combinations. Connectors make it possible to deploy generic business rules in a specific context. Imagine our application implements a `checkout`-method that iterates over all purchased products and returns the total price. Figure 3 illustrates the `ChristmasDiscountDeployment`-Connector that deploys the `ChristmasBR`-rule upon this `checkout`-method.

```

1 connector ChristmasDiscountDeployment {
2     ChristmasBR.BRPriceDiscountHook discount = new
3     ChristmasBR.BRPriceDiscountHook(Float CheckOut.CheckOut(Customer));
4
5     discount.setDiscount(new Float(0.05));
6     discount.replace();
7 }

```

**Figure 3: Deployment of the Christmas business rule**

The Connector of Figure 3 initialises the `BRPriceDiscount`-hook with the `checkOut`-method defined in the `CheckOut`-class (lines 2 to 3). After initialising this hook, the `ChristmasDiscountDeployment`-Connector specifies the exact discount (line 5) and the execution of the `replace` behaviour method (line 6). Consequently, the deployment of this Connector has the following implication: apply a discount of 5% on the total price when a customer checks out during the Christmas period.

## 4 JASCO FOR BUSINESS RULES

### 4.1 Explicit Connectors

As mentioned before, one of the main advantages of the use of JAsCo is the separation and encapsulation of the deployment details in a new Connector construct. For achieving the decoupling of the business rule link, the abstract logic for the application of a business rule is specified by using a generic hook defined in the Aspect Bean. This way, the crosscutting code remains independent from the details of the concrete deployments and is encapsulated in the Connectors.

The example illustrated in Figure 3, specifies the deployment of the application logic of the `ChristmasBR` whenever the `checkout` method is executed. Suppose that the business requirements change and the `ChristmasBR` should be applied only on a specialised customer such as an employee of the firm. This requirement can easily be achieved by creating another Connector that instantiates the same `BRPriceDiscountHook`, providing the `EmployeeCustomer` as a parameter for the `checkout` method. This way, the specification of this new deployment is encapsulated in the new Connector without affecting the previous abstract definition.

Another advantage of having explicit Connectors is the possibility to group together the deployment details of logically related business rules. This advantage is illustrated by introducing the following example. Suppose customers must be classified by considering them as frequent or not frequent. To achieve this, a new business rule is specified: if the customer purchased more than 10 items, then the customer is frequent. Figure 4

shows the implementation of this business rule as an Aspect Bean.

```
1  class FrequentCustomer {
2      public boolean checkFrequentCustomerCondition(Customer customer) {
3          if (!FrequentCustomers.isFrequentCustomer(customer)) {
4              int noProducts = customer.getAccount().getBoughtProducts();
5              return noProducts > 10;
6          } else return false;
7      }
8
9      hook FrequentCustomerHook {
10         FrequentCustomerHook(Float method(Customer customer)) {
11             execute(method);
12         }
13         isApplicable() {
14             return checkFrequentCustomerCondition(customer);
15         }
16         after() {
17             FrequentCustomers.addFrequentCustomer(customer);
18         }
19     }
20 }
```

**Figure 4: The FrequentCustomer aspect**

Now consider a new business rule for the price personalisation that makes use of this new concept of customer frequency: if the customer is frequent, then apply a 5% discount. The Aspect Bean `FrequentCustomerBR` that implements this rule (Figure 5) extends the `BRPriceDiscount-Aspect` Bean as it is a rule for price personalisation.

```
1  class FrequentCustomerBR extends BRPriceDiscount {
2      public boolean discountCondition(Customer customer) {
3          return FrequentCustomers.isFrequentCustomer(customer);
4      }
5  }
```

**Figure 5: The FrequentCustomer business rule**

In Figure 6, the `FrequentCustomers` class is introduced for holding the frequent customer information that is shared among the two business rules.

```
1  class FrequentCustomers {
2      private static Vector customers = new Vector();
3      public static void addFrequentCustomer(Customer customer) {
4          customers.add(customer);
5      }
6      public static boolean isFrequentCustomer(Customer customer) {
7          return customers.contains(customer);
8      }
9  }
```

**Figure 6: The FrequentCustomer-class**



Both rules are logically related, because they specify business considerations about customer frequency. As a result, only one Connector needs to be defined to gather the concrete information about the deployment of both rules. Another advantage of having the deployment information in the same Connector is that the order in which the application of the rules should be triggered can be controlled by explicitly invoking the application of the rules in the desired order. Figure 7 illustrates the implementation of the Connector for the deployment of both rules.

```

1  connector FrequentCustomerDiscountDeployment {
2
3      FrequentCustomerBR.BRPriceDiscountHook discount = new
4          FrequentCustomerBR.BRPriceDiscountHook(
5              Float CheckOut.CheckOut(Customer));
6      discount.setDiscount(0.05);
7
8      FrequentCustomer.FrequentCustomerHook frequent = new
9          FrequentCustomerBR.FrequentCustomerHook(
10         FloatCheckOut.CheckOut(Customer));
11
12     discount.replace();
13     frequent.after();
14 }

```

Instantiating two related hooks

**Figure 7: The deployment of the Frequent Customer business rule**

## 4.2 Precedence and Combination Strategies

When several business rules are deployed within a single software system, it is possible that these rules influence each other's execution. This problem is a well-known issue in AOP, and is identified as the *feature interaction problem* [20]. To solve this problem, the JAsCo language provides a powerful, reusable and extensive system for specifying the precedence and the combination of aspects.

### 4.2.1 Precedence Strategies

The JAsCo language allows arranging the execution of a set of business rules, by explicitly specifying the desired sequence in the Connector. Whenever two or more hooks interfere, the order in which their behaviour must be executed is derived from the Connector. Figure 8 illustrates the deployment of a business strategy where the Christmas discount is given prior to the frequent customer discount.

```


1  connector ChristmasFrequentCustomerDiscountDeployment {
2
3      ChristmasBR.BRPriceDiscountHook christmasDiscount = new
4          ChristmasBR.BRPriceDiscountHook(
5              Float CheckOut.CheckOut(Customer));
6      christmasDiscount.setDiscount2(0.10);

```

```

7
8   FrequentCustomerBR.BRPriceDiscountHook frequentDiscount = new
9     FrequentCustomerBR.BRPriceDiscountHook(
10      Float CheckOut.CheckOut(Customer));
11   frequentDiscount.setDiscount(0.05);
12
13   FrequentCustomerBR.FrequentCustomerHook frequentChecker = new
14     FrequentCustomerBR.FrequentCustomerHook(
15      Float CheckOut.CheckOut(Customer));
16
17   christmasDiscount.replace();
18   frequentDiscount.replace();
19
20   frequentChecker.after();
21 }

```



**Figure 8: Additive deployment of discount business rules**

## 4.2.2 Combination Strategies

Being able to specify the sequence in which the various business rules are executed is in many cases not sufficient. Some business strategies require more advanced techniques to specify the combination of the various business rules that are deployed within the system. In the previous section for instance, an additive discount strategy is employed. However, the business policy could specify that only one discount is offered for a given product: if somebody buys an item during the Christmas period, the frequent customer discount is not applicable. The JAsCo language provides a solution to be able to specify this kind of advanced aspect-combinations, by providing a mechanism called combination strategies. A combination strategy acts like a kind of filter that validates the list of applicable hooks, which are obtained at run time. Each specific combination strategy implements the `CombinationStrategy`-interface introduced in Figure 9. The interface itself only specifies the `validateCombinations`-method, which is used to describe the specific logic of a combination strategy. This mechanism of combination strategies allows maximum flexibility, as user-defined relationships between the various aspects can be specified.

```

1 public interface CombinationStrategy {
2     public HookList validateCombinations(HookList aHookList);
3 }

```

**Figure 9: The `CombinationStrategy` interface**

The exclude combination strategy illustrated in Figure 10 specifies a combination strategy where the behaviour of hook *B* cannot be executed whenever hook *A* is encountered. This combination strategy can be used to specify the relationship between the Christmas and the frequent customer discount business rules.

```

1  class ExcludeCombinationStrategy implements CombinationStrategy {
2      private Object A;
3      private Object B;
4      public ExcludeCombinationStrategy(Object hookA, Object hookB) {
5          A = hookA;
6          B = hookB;
7      }
8      public HookList validateCombinations(HookList aHookList) {
9          if (aHookList.contains(A)) {
10             aHookList.remove(B);
11         }
12         return aHookList;
13     }
14 }

```

**Figure 10: The Exclude CombinationStrategy**

The Connector illustrated in Figure 11 deploys the Christmas and the frequent customer discount business rule. Both business rules are initialised with a specific context, and the execution of their behaviour methods is specified. The Connector however also specifies an exclude combination strategy between both business rules (lines 13 to 15). As a result, whenever the Christmas discount is applied, the behaviour of the frequent customer business rule is ignored.

```

1  connector ChristmasFrequentCustomerDiscountDeployment {
2
3      ChristmasBR.BRPriceDiscountHook christmasDiscount = new
4          ChristmasBR.BRPriceDiscountHook(
5          Float CheckOut.CheckOut(Customer));
6      christmasDiscount.setDiscount2(0.10);
7
8      FrequentCustomerBR.BRPriceDiscountHook frequentDiscount = new
9          FrequentCustomerBR.BRPriceDiscountHook(
10         Float CheckOut.CheckOut(Customer));
11     frequentDiscount.setDiscount(0.05);
12
13     ExcludeCombinationStrategy strategy = new
14         ExcludeCombinationStrategy(christmasDiscount, frequentDiscount);
15     addCombinationStrategy(strategy);
16
17     FrequentCustomerBR.FrequentCustomerHook frequentChecker = new
18         FrequentCustomerBR.FrequentCustomerHook(
19         Float CheckOut.CheckOut(Customer));
20
21     christmasdiscount.replace();
22     frequentdiscount.replace();
23
24     frequentChecker.after();
25 }

```

*Specifying an exclude combination strategy*

**Figure 11: Exclusive deployment of discount business rules**

### 4.3 Controlled Instantiation, Initialisation and Execution of Aspects

Most aspect-oriented technologies do not allow sophisticated control for instantiating, initialising and executing aspects, as this is done implicitly when the aspect is woven into the core of the base application. The JAsCo system improves upon these techniques, as the instantiation of an aspect with a specific context is described explicitly in the Connector. As a result, every instantiated aspect can be accessed as being a first class entity. This allows initialising each aspect instance with some specific properties. Considering the business rules environment, this is a vital contribution, as it allows fine-tuning general-purpose business rules to conform to the specific business requirements. Also, the execution of the behaviour of the business rules is specified explicitly in the Connector, which allows even more fine-grained control.

### 4.4 Dynamic Reconfiguration of Business rules

Business rules tend to evolve continuously in comparison to the core functionality of the system. Some business rules, such as the Christmas discount rule introduced in Section 3, are only obligatory during a certain period of the year. Other business rules need to be adapted regularly to be able to fulfil new business requirements. Consequently, it should be possible to add, edit and remove business rules at Connector. Current AOP technologies however, do not allow easy management of business rules, as the deployment of an aspect within the system is rather static. This is mainly because an aspect loses its identity when it is woven into the base-application. JAsCo solves this issue, by also providing a run-time separation between the aspects and the base implementation of the system. This way, JAsCo aspects remain first class entities when they are deployed and their logic is not weld together with the base functionality of the application. This property of the JAsCo system is a valuable concept in the business rules environment, as this run-time separation, together with the new component model, allows dynamic reconfiguration of business rules, without the need to shut down business-critical applications.

## 5 RELATED WORK

Several approaches exist that deal with business rules at the implementation level and are compatible with object-oriented software engineering. *Business Rule Beans* [23] provides a Java framework to externalise business rules and trigger them manually and explicitly in the core application. *CommonRules* [13] is a Java library implementing *Situated Courteous Logic* as a rule-based forward chaining engine. Business rules are also

externalised through the use of rule sets that can be called from the core application. The Situated Courteous Logic allows for the specification of business rule combinations (priority and mutual exclusion) and attached procedures. A whole range of object-oriented patterns are defined for supporting business rules, such as the *Rule Object Pattern* [1], *Patterns for Personalisation* [21], and *Rule Patterns* [15]. All of these approaches have in common that linking the business rules requires manually adapting the source code of the core application in different places in order to apply the business rules. As explained earlier, this results in crosscutting code, which cannot be encapsulated nor reused.

Aspect-Oriented approaches are originally conceived with low-level implementation aspects in mind, such as synchronisation, error handling and logging. However, Tarr et al. also apply their idea of *Multi-Dimensional Separation of Concerns* and their tool HyperJ [19][27] to other kinds of concerns like business rules. In their approach, business rules can be encapsulated in different hyperslices, which are their modularization mechanism for crosscutting concerns. Hyperslices are loosely coupled with the base model, which implies that the business rules they encapsulate are reusable in different contexts. In this approach it is possible to specify a separate module (hypermodule) to encapsulate the details of how the business rules are linked to the core application. However, not much support for hyperslice relations is provided, limiting the combination of business rules. Moreover, mapping concerns is done statically, by matching structural units present in different hyperslices. This characteristic does not allow the connection of business rules to core application events that depend on run-time properties, one of the desired requirements we pursue.

## 6 CONCLUSIONS

The main goal of this research consists of realizing independent, reusable and manageable business rules at the implementation level of object-oriented software applications. In order to achieve this we propose to use aspect-oriented ideas to link the business rules to the core application. In a previous attempt, we used AspectJ as a concrete AOP technology and identified several problems. In this paper, we show that JAsCo is able to improve on AspectJ for representing business rules on several essential points. First of all, JAsCo allows specifying reusable business rules that can be instantiated to fit the application at hand. Secondly, the Connector concept of JAsCo allows controlling the instantiation and initialisation of the business rules. An additional advantage of the Connector is that it allows specifying and managing more advanced and fine-grained business rule combinations than in AspectJ. Last but not least, JAsCo allows run-time application and removal of business rules which is an essential property in this context. On the other hand, some considerations need to be

taken into account. JAsCo is a rather new AOP language whereas AspectJ is already mature and applied to large industrial case studies. In addition, AspectJ offers more advanced join point expressions than JAsCo.

This paper takes an important step in bridging the gap between business rule specification and implementation. The use of AOP and in particular JAsCo enables us to maintain the modularity conceived at the conceptual level to the implementation level. However, in this work the representation of the business rules themselves changes from a conceptual *if-then* format to objects. The reason is that this allows us to concentrate on the business rule link and it also facilitates the use of existing AOP approaches such as AspectJ and JAsCo since they extend Java. A continuation of this work is to consider a more suitable representation for business rules, such as a rule-based language [13], in order to minimise transition from specification to implementation even more.

## 7 REFERENCES

- [1] A. Arsanjani. Rule object 2001: A pattern language for adaptive and scalable business rule construction.
- [2] Arsanjani and J. Alpignì. Using grammar-oriented object design to seamlessly map business models to component-based software architectures. In International Symposium of Modelling and Simulation, Pittsburgh, USA, pages 186--191, 2001.
- [3] Aspect-Oriented Software Development. <http://www.aosd.net/>
- [4] Bergmans, L. and Aksit, M. Composing Crosscutting Concerns Using Composition Filters. Communications of the ACM, Vol. 44, No. 10, pp. 51-57, October 2001.
- [5] The Business Rules Group. Defining Business Rules: What Are They Really?, <http://www.businessrulesgroup.org/>, July 2000.
- [6] M. A. Cibrán. Using aspect-oriented programming for connecting and configuring decoupled business rules in object-oriented applications. Master Thesis, Vrije Universiteit Brussel, Belgium, 2002.
- [7] M. A. Cibrán and M. D'Hondt. Aspect-Oriented Programming for Connecting Business Rules. 6<sup>th</sup> International Conference on Business Information Systems, Colorado Springs, USA, 2003. (submitted)
- [8] Communications of the ACM. Aspect-Oriented Software Development, October 2001.
- [9] Communications of the ACM. The Adaptive Web, June 2002.
- [10] C. Date. What not How: The Business Rules Approach to Application Development. Addison-Wesley Publishing Company, 2000.
- [11] M. D'Hondt, W. De Meuter and R. Wuyts. Using Reflective Logic Programming to Describe Domain Knowledge as an Aspect. In Proceedings of the First Symposium on Generative and Component-Based Software Engineering. Erfurt, Germany, 1999.
- [12] M. D'Hondt and T. D'Hondt. Is Domain Knowledge an Aspect? ECOOP '99, Workshop on Aspect-Oriented Programming. Lisbon, Portugal, 1999.
- [13] B. N. Grosf, Y. Kabbaj, T. C. Poon, M. Ghande, and et al. Semantic Web Enabling Technology (SWEET). <http://ebusiness.mit.edu/bgrosf/>
- [14] B. N. Grosf, Y. Labrou, and H. Y. Chan. A declarative approach to business rules in contracts: courteous logic programs in xml. In Proceedings of the first ACM conference on Electronic commerce, pages 68--77. ACM Press, 1999.
- [15] G. Kappel, S. Rausch-Schott, W. Retschitzegger, and M. Sakkinen. From rules to rule patterns. In Conference on Advanced Information Systems Engineering, pages 99--115, 1996.
- [16] G. Kiczales, E. Hilsdale, J. Hugunin, M. Kersen, J. Palm, and W. G. Griswold. An overview of AspectJ. In Proceedings European Conference on Object-Oriented Programming, volume 2072 of Lecture Notes in Computer Science, pages 327--353, Berlin, Heidelberg, and New York, 2001. Springer-Verlag.
- [17] Lieberherr, K., Lorenz, D. and Mezini, M. Programming with Aspectual Components. Technical Report, NU-CCS-99-01, March 1999. Available at: <http://www.ccs.neu.edu/research/demeter/biblio/aspectual-comps.html>.
- [18] Lieberherr, K., Orleans, D., Ovlinger, J. Aspect-oriented programming with adaptive methods. Communications of the ACM, Vol. 44, No. 10, pp. 39-41, October 2001.
- [19] H. Ossher and P. Tarr. Using multidimensional separation of concerns to (re)shape evolving software. Communications of the ACM, 44(10):43--50, Oct. 2001.
- [20] E. Pulvermueller, A. Speck, M. D'Hondt, W. De Meuter, and J. O. Coplien. Report from the ECOOP 2001 Workshop on Feature Interaction in Composed Systems. In Workshop Reader of the 15<sup>th</sup> European Conference on Object-Oriented Programming (ECOOP '01). Springer-Verlag, 2001. Program available at: <http://www.info.uni-arlstruhe.de/~pulvermu/workshops/ecoop2001/>

- [21] G. Rossi, A. Fortier, J. Cappi, and D. Schwabe. Seamless personalization of e-commerce applications. In 2nd International Workshop on Conceptual Modeling Approaches for e-Business at the 20th International Conference on Conceptual Modeling, Yokohama, Japan, 2001.
- [22] G. Rossi, D. Schwabe, and R. Guimaraes. Designing personalized web applications. In World Wide Web, pages 275--284, 2001.
- [23] I. Rouvellou, L. Degenaro, K. Rasmus, D. Ehnebuske, and B. McKee. Extending business objects with business rules. In 33rd International Conference on Technology of Object-Oriented Languages and Systems ( TOOLS Europe 2000), Mont Saint-Michel - St-Malo, France, pages 238--249, 2000.
- [24] Schreiber, G., Akkermans, H., Anjewierden, A., de Hoog, R., Shadbolt, N., Van de Velde, W. and Wielinga, B. Knowledge Engineering and Management: The CommonKADS Methodology. MIT Press (2000).
- [25] C. Szyperski. Component software: Beyond Object-oriented programming. Addison-Wesley, 1998.
- [26] D. Suvée and W. Vanderperren. JAsCo: an Aspect-Oriented approach tailored for Component Based Software Development. In Proceedings of the 2<sup>nd</sup> International Conference on Aspect-Oriented Software Development, Boston, USA, 2003.
- [27] Peri Tarr, Harold Ossher, William Harrison, and Stanley M. Sutton, Jr. N degrees of separation: Multi-dimensional separation of concerns. In Proceedings of the 1999 International Conference on Software Engineering, pages 107-119. IEEE Computer Society Press / ACM Press, 1999.
- [28] B. von Halle. Business Rules Applied. Wiley, 2001.